In[•]:=
```
<< KnotTheory`
```

```
Loading KnotTheory` version of October 29, 2024, 10:29:52.1301.
Read more at http://katlas.org/wiki/KnotTheory.
```

In[•]:=
```
PD[GST48] = PD[X[1, 15, 2, 14], X[29, 2, 30, 3], X[40, 4, 41, 3],
   X[4, 44, 5, 43], X[5, 26, 6, 27], X[95, 7, 96, 6], X[7, 1, 8, 96], X[8, 14, 9, 13],
   X[28, 9, 29, 10], X[41, 11, 42, 10], X[11, 43, 12, 42], X[12, 27, 13, 28],
   X[15, 31, 16, 30], X[61, 16, 62, 17], X[72, 17, 73, 18], X[83, 18, 84, 19],
   X[34, 20, 35, 19], X[20, 89, 21, 90], X[92, 21, 93, 22], X[22, 79, 23, 80],
   X[23, 68, 24, 69], X[24, 57, 25, 58], X[56, 25, 57, 26], X[31, 63, 32, 62],
   X[32, 74, 33, 73], X[33, 85, 34, 84], X[35, 50, 36, 51], X[81, 37, 82, 36],
   X[70, 38, 71, 37], X[59, 39, 60, 38], X[54, 39, 55, 40], X[55, 45, 56, 44],
   X[45, 59, 46, 58], X[46, 70, 47, 69], X[47, 81, 48, 80], X[91, 49, 92, 48],
   X[49, 91, 50, 90], X[82, 52, 83, 51], X[71, 53, 72, 52], X[60, 54, 61, 53],
   X[74, 63, 75, 64], X[85, 64, 86, 65], X[65, 76, 66, 77], X[66, 87, 67, 88],
   X[94, 67, 95, 68], X[86, 75, 87, 76], X[77, 88, 78, 89], X[93, 78, 94, 79]];
```

In[•]:=
```
SetAttributes[p, Orderless]
```

In[•]:=
```
K = Knot[10, 165];
pd = PD[K]
```

```
KnotTheory: Loading precomputed data in PD4Knots`.
```

Out[•]=
```
PD[X[1, 6, 2, 7], X[7, 18, 8, 19], X[3, 9, 4, 8], X[17, 3, 18, 2], X[5, 15, 6, 14],
 X[9, 17, 10, 16], X[15, 11, 16, 10], X[11, 5, 12, 4], X[20, 14, 1, 13], X[12, 20, 13, 19]]
```

In[•]:=
```
t1 = Times @@ pd /. X[i_, j_, k_, l_] :> A p[i, j] p[k, l] + A^-1 p[i, l] p[j, k]
```

Out[•]=

$$\left(\frac{p[1,7]\,p[2,6]}{A}+A\,p[1,6]\,p[2,7]\right)\left(A\,p[2,18]\,p[3,17]+\frac{p[2,17]\,p[3,18]}{A}\right)$$
$$\left(A\,p[3,9]\,p[4,8]+\frac{p[3,8]\,p[4,9]}{A}\right)\left(A\,p[4,12]\,p[5,11]+\frac{p[4,11]\,p[5,12]}{A}\right)$$
$$\left(A\,p[5,15]\,p[6,14]+\frac{p[5,14]\,p[6,15]}{A}\right)\left(\frac{p[7,19]\,p[8,18]}{A}+A\,p[7,18]\,p[8,19]\right)$$
$$\left(A\,p[9,17]\,p[10,16]+\frac{p[9,16]\,p[10,17]}{A}\right)\left(A\,p[10,16]\,p[11,15]+\frac{p[10,15]\,p[11,16]}{A}\right)$$
$$\left(A\,p[12,20]\,p[13,19]+\frac{p[12,19]\,p[13,20]}{A}\right)\left(\frac{p[1,14]\,p[13,20]}{A}+A\,p[1,13]\,p[14,20]\right)$$

```
In[•]:= t2 = Expand[t1]
```

Out[•]=

$$A^4\,p[1,7]\,p[1,14]\,p[2,6]\,p[2,18]\,p[3,9]\,p[3,17]\,p[4,8]\,p[4,12]\,p[5,11]\,p[5,15]\,p[6,14]\,p[7,19]\,p[8,18]\,p[9,17]\,p[10,16]^2\,p[11,15]\,p[12,20]\,p[13,19]\,p[13,20] + A^6\,p[1,6]\,p[1,14]\,p[2,7]\,p[2,18]\ \langle\!\langle 11 \rangle\!\rangle\ p[11,15]\,p[12,20]\,p[13,19]\,p[13,20] + \langle\!\langle 1020 \rangle\!\rangle + \frac{\langle\!\langle 1 \rangle\!\rangle}{A^6} + \frac{p[1,6]\,p[1,13]\,p[2,7]\,p[2,17]\,p[3,8]\ \langle\!\langle 11 \rangle\!\rangle\ p[11,16]\,p[12,19]\,p[13,20]\,p[14,20]}{A^4}$$

Full expression not available (original memory size: 2.1 MB)

```
In[•]:= t3 = t2 //. p[i_, j_] p[j_, k_] :> p[i, k]
```

```
In[•]:= t4 = t3 /. {p[_, _]^2 → d}
```

Out[•]=

$$\frac{70\,d}{A^2} + 31\,A^2\,d + 171\,d^2 + \frac{87\,d^2}{A^4} + 42\,A^4\,d^2 + \frac{42\,d^3}{A^6} + \frac{132\,d^3}{A^2} + 147\,A^2\,d^3 + 26\,A^6\,d^3 + 80\,d^4 + \frac{10\,d^4}{A^8} + \frac{33\,d^4}{A^4} + 70\,A^4\,d^4 + 8\,A^8\,d^4 + \frac{d^5}{A^{10}} + \frac{3\,d^5}{A^6} + \frac{8\,d^5}{A^2} + 32\,A^2\,d^5 + 18\,A^6\,d^5 + A^{10}\,d^5 + d^6 + 8\,A^4\,d^6 + 2\,A^8\,d^6 + A^6\,d^7$$

```
In[•]:= t5 = Expand[t4 /. d → -A^2 - A^-2]
```

Out[•]=

$$-1 - \frac{1}{A^{20}} + \frac{2}{A^{16}} - \frac{1}{A^{12}} + \frac{2}{A^8} - \frac{1}{A^4} - 2\,A^8 + 2\,A^{12} - 2\,A^{16}$$

```
In[•]:= SKB[K_] := Module[{pd, t1, t2, t3, t4},
  pd = PD[K];
  t1 = Times @@ pd /. X[i_, j_, k_, l_] :> A p[i, j] p[k, l] + A^-1 p[i, l] p[j, k];
  t2 = Expand[t1];
  t3 = t2 //. p[i_, j_] p[j_, k_] :> p[i, k];
  t4 = t3 /. {p[_, _]^2 → d};
  Expand[t4 /. d → -A^2 - A^-2]
 ]
```

```
In[•]:= SKB[Knot[4, 1]]
```

Out[•]=

$$-\frac{1}{A^{10}} - A^{10}$$

```
In[•]:= SKB[Knot[8, 17]]
```

Out[•]=

$$-\frac{1}{A^{18}} + \frac{2}{A^{14}} - \frac{2}{A^{10}} + \frac{1}{A^6} - \frac{1}{A^2} - A^2 + A^6 - 2\,A^{10} + 2\,A^{14} - A^{18}$$

```
In[•]:= (*SKB[GST48]*)
```

```
In[•]:= K = Knot[8, 17]
```

Out[•]=

Knot[8, 17]

```
In[•]:= pd = List @@ PD[K]
        front = {}
        kb = 1
Out[•]=
        {X[6, 2, 7, 1], X[14, 8, 15, 7], X[8, 3, 9, 4], X[2, 13, 3, 14],
         X[12, 5, 13, 6], X[4, 9, 5, 10], X[16, 12, 1, 11], X[10, 16, 11, 15]}
Out[•]=
        {}
Out[•]=
        1

In[•]:= u[X[i_, j_, k_, l_]] := Length[front ⋂ {i, j, k, l}]

In[•]:= u /@ pd
Out[•]=
        {0, 0, 0, 0, 0, 0, 0, 0}

In[•]:= x = RandomChoice[MaximalBy[pd, u]]
Out[•]=
        X[10, 16, 11, 15]

In[•]:= t1 = Expand[kb (x /. X[i_, j_, k_, l_] :↦ A p[i, j] p[k, l] + A^-1 p[i, l] p[j, k])] //.
          p[i_, j_] p[j_, k_] :↦ p[i, k]
Out[•]=
        A p[10, 16] p[11, 15] + (p[10, 15] p[11, 16]) / A

In[•]:= t2 = t1 /. {p[_, _]^2 → d}
Out[•]=
        A p[10, 16] p[11, 15] + (p[10, 15] p[11, 16]) / A

In[•]:= kb = Expand[t2 /. d → -A^2 - A^-2]
Out[•]=
        A p[10, 16] p[11, 15] + (p[10, 15] p[11, 16]) / A

In[•]:= pd = DeleteCases[pd, x]
Out[•]=
        {X[6, 2, 7, 1], X[14, 8, 15, 7], X[8, 3, 9, 4],
         X[2, 13, 3, 14], X[12, 5, 13, 6], X[4, 9, 5, 10], X[16, 12, 1, 11]}

In[•]:= front = Complement[front ⋃ List @@ x, front ⋂ List @@ x]
Out[•]=
        {10, 11, 15, 16}
```

```
In[•]:= KB[K_] := Module[{t1, t2, pd, front, kb, u, x},
          pd = List @@ PD[K];
          front = {};
          kb = 1;
          u[X[i_, j_, k_, l_]] := Length[front ⋂ {i, j, k, l}];
          While[pd =!= {},
           x = RandomChoice[MaximalBy[pd, u]];
           t1 = Expand[kb (x /. X[i_, j_, k_, l_] :↦ A p[i, j] p[k, l] + A^-1 p[i, l] p[j, k])] //.
             p[i_, j_] p[j_, k_] :↦ p[i, k];
           t2 = t1 /. {p[_, _]^2 → d};
           kb = Expand[t2 /. d → -A^2 - A^-2];
           pd = DeleteCases[pd, x];
           front = Complement[front ⋃ List @@ x, front ⋂ List @@ x]
          ];
          kb
         ]
```

```
In[•]:= KB[Knot[8, 17]]
```

```
Out[•]=
```

$$-\frac{1}{A^{18}}+\frac{2}{A^{14}}-\frac{2}{A^{10}}+\frac{1}{A^6}-\frac{1}{A^2}-A^2+A^6-2\,A^{10}+2\,A^{14}-A^{18}$$

```
In[•]:= KB[GST48]
```

```
Out[•]=
```

$$\frac{1}{A^{60}}-\frac{2}{A^{56}}+\frac{1}{A^{52}}+\frac{1}{A^{48}}-\frac{2}{A^{44}}-\frac{1}{A^{24}}+\frac{2}{A^{20}}+\frac{1}{A^{16}}-\frac{2}{A^{12}}+\frac{4}{A^8}-\frac{2}{A^4}-2\,A^8-2\,A^{16}+A^{20}-A^{28}+A^{36}$$